// File: PrideIconsAuction.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.11;

import "Ownable.sol";
import "PrideIcons.sol";

contract PrideIconsAuction is Ownable {
    uint16 internal constant SECONDS_IN_HOUR = 3600;
    uint8 internal constant DEFAULT_INTERVAL_HOURS = 24;
    uint8 internal constant DEFAULT_BIDS_IN_PARALLEL = 3;

    uint256 public minimumBid;

    struct BidOpenTimes {
        // Timestamps - when bidding open / closed for given token.
        uint256 biddingOpen;
        uint256 biddingClosed;
    }
    mapping(uint8 => BidOpenTimes) public bidTimestamps;

    struct Bid {
        address walletBidder;
        address walletRecipient;
        uint256 bidAmount;
    }
    // Token ID => all bids
    mapping(uint8 => Bid[]) public bidHistory;

    // Token ID => auction cancelled / not cancelled
    mapping(uint8 => bool) public cancelledAuctions;

    // Paper wallets
    mapping(address => bool) private paperWallets;

    bool public locked;

    PrideIcons public prideIcons;

    // @VisibleForTesting
    address internal vaultAddress;

    event BidPlaced(
        uint8 indexed tokenId,
        address bidder,
        address recipient,
        uint256 amount
    );
    event AuctionEnded(
        uint8 indexed tokenId,
        address bidder,
        address recipient,
        uint256 amount
    );
    event AuctionCancelled(uint8 indexed tokenId, address bidder);

    constructor(
        address _vaultAddress,
        uint256 _minimumBid,
        address _prideIconsAddress,
        uint256 _timestamp,
        address[] memory _paperWallets
    ) {
        require(_vaultAddress != address(0), "NULL_ADDRESS");
        vaultAddress = _vaultAddress;
        minimumBid = _minimumBid;
        prideIcons = PrideIcons(_prideIconsAddress);
        initializeTimestamps(
            _timestamp,
            DEFAULT_INTERVAL_HOURS,
            DEFAULT_BIDS_IN_PARALLEL
        );
        initializePaperWallets(_paperWallets);
    }

    modifier notLocked() {
        require(!locked, "CONTRACT_METADATA_METHODS_ARE_LOCKED");
        _;
    }

    modifier onlyPaperOrOwner() {
        require(
            paperWallets[msg.sender] || owner() == msg.sender,
            "ONLY_PAPER_WALLETS_OR_OWNER_ALLOWED"
        );
        _;
    }

    // Method for Paper.xyz (sends on behalf of another wallet)
    function placeBid(uint8 tokenId, address recipient)
        external
        payable
        onlyPaperOrOwner
    {
        _placeBid(tokenId, recipient);
    }

    function placeBid(uint8 tokenId) external payable {
        _placeBid(tokenId, msg.sender);
    }

    function _placeBid(uint8 tokenId, address recipient) private notLocked {
        Bid memory currentTopBid = getTopBid(tokenId);
        require(
            block.timestamp >= bidTimestamps[tokenId].biddingOpen &&
                block.timestamp <= bidTimestamps[tokenId].biddingClosed,
            "BIDDING_FOR_TOKEN_CLOSED"
        );
        require(!cancelledAuctions[tokenId], "AUCTION_CANCELLED");

        Bid memory latestBid = Bid(msg.sender, recipient, msg.value);
        if (currentTopBid.bidAmount == 0) {
            require(latestBid.bidAmount >= minimumBid, "BID_UNDER_MINIMUM");
        } else {
            require(
                latestBid.bidAmount > currentTopBid.bidAmount,
                "BID_TOO_LOW"
            );
            payable(currentTopBid.walletBidder).transfer(
                currentTopBid.bidAmount
            );
        }
        bidHistory[tokenId].push(latestBid);
        emit BidPlaced(
            tokenId,
            latestBid.walletBidder,
            latestBid.walletRecipient,
            latestBid.bidAmount
        );
    }

    function endAuction(uint8 tokenId) external notLocked {
        Bid memory currentTopBid = getTopBid(tokenId);
        require(!cancelledAuctions[tokenId], "AUCTION_CANCELLED");
        require(
            msg.sender == currentTopBid.walletBidder ||
                paperWallets[msg.sender] ||
                msg.sender == owner(),
            "ONLY_BIDDER_OR_PAPER_OR_OWNER_ALLOWED"
        );
        require(
            block.timestamp >= bidTimestamps[tokenId].biddingClosed,
            "BIDDING_NOT_OVER"
        );
        require(address(prideIcons) != address(0), "NULL_CONTRACT_ADDRESS");
        require(currentTopBid.walletRecipient != address(0), "NO_WINNERS");
        prideIcons.mintPlatinum(tokenId, currentTopBid.walletRecipient);

        emit AuctionEnded(
            tokenId,
            currentTopBid.walletBidder,
            currentTopBid.walletRecipient,
            currentTopBid.bidAmount
        );
    }

    function cancelAuction(uint8 tokenId) external onlyPaperOrOwner notLocked {
        Bid memory currentTopBid = getTopBid(tokenId);
        require(!cancelledAuctions[tokenId], "ALREADY_CANCELLED");
        require(currentTopBid.bidAmount != 0, "MUST_HAVE_BID");
        require(currentTopBid.walletBidder != address(0), "MUST_HAVE_BIDDER");
        require(
            paperWallets[currentTopBid.walletBidder] || msg.sender == owner(),
            "ONLY_PAPER_BIDDER_OR_OWNER_ALLOWED"
        );
        require(address(prideIcons) != address(0), "NULL_CONTRACT_ADDRESS");
        require(
            !prideIcons.hasMintedPlatinum(tokenId),
            "AUCTION_ALREADY_ENDED"
        );

        // Pay back bid to top bidder.
        payable(currentTopBid.walletBidder).transfer(currentTopBid.bidAmount);

        // Cancel auction.
        cancelledAuctions[tokenId] = true;

        emit AuctionCancelled(tokenId, currentTopBid.walletBidder);
    }

    function getBidHistory(uint8 tokenId) external view returns (Bid[] memory) {
        return bidHistory[tokenId];
    }

    function initializeTimestamps(
        uint256 firstBidStartTime,
        uint256 intervalDurationHours,
        uint8 bidsInParallel
    ) public onlyOwner {
        require(address(prideIcons) != address(0), "NULL_CONTRACT_ADDRESS");
        uint16 platinumSupply = prideIcons.platinumSupply();

        for (uint8 i = 1; i <= platinumSupply; i++) {
            for (uint8 j = 0; j < bidsInParallel; j++) {
                // Assign biddingOpen and biddingClosed fields
                uint8 index = i + j + (bidsInParallel - 1) * (i - 1);
                if (index > platinumSupply) return;
                bidTimestamps[index] = BidOpenTimes(
                    firstBidStartTime +
                        ((i - 1) * intervalDurationHours * SECONDS_IN_HOUR),
                    firstBidStartTime +
                        (i * intervalDurationHours * SECONDS_IN_HOUR)
                );
            }
        }
    }

    function getTopBid(uint8 tokenId) public view returns (Bid memory) {
        Bid[] memory bids = bidHistory[tokenId];
        return
            bids.length == 0
                ? Bid(address(0), address(0), 0)
                : bids[bids.length - 1];
    }

    // virtual for tests
    function initializePaperWallets(address[] memory wallets)
        public
        virtual
        onlyOwner
    {
        for (uint8 i = 0; i < wallets.length; i++) {
            paperWallets[wallets[i]] = true;
        }
    }

    function withdrawAll() external onlyOwner {
        payable(vaultAddress).transfer(address(this).balance);
    }

    // Ad hoc modification of bidding times for a specific tokens.
    function modifyTimeStampForTokens(
        uint8[] calldata tokenIds,
        BidOpenTimes[] calldata biddingTimes
    ) external onlyOwner {
        require(tokenIds.length == biddingTimes.length, "UNEQUAL_LENGTH");
        for (uint8 i = 0; i < tokenIds.length; i++) {
            bidTimestamps[tokenIds[i]] = biddingTimes[i];
        }
    }

    function setVaultAddress(address newVault) external onlyOwner {
        require(newVault != address(0), "NULL_ADDRESS");
        vaultAddress = newVault;
    }

    function setMinimumBid(uint256 _minimumBid) external onlyOwner {
        minimumBid = _minimumBid;
    }

    function assignNftContractAddress(address contractAddress)
        external
        onlyOwner
    {
        require(contractAddress != address(0), "NULL_ADDRESS");
        prideIcons = PrideIcons(contractAddress);
    }

    function lockAuction() external onlyOwner {
        locked = true;
    }

    function unlockAuction() external onlyOwner {
        locked = false;
    }

    struct CancelledAuctionMap {
        uint8 tokenId;
        bool cancelled;
    }
    struct BidHistoryMap {
        uint8 tokenId;
        Bid[] bids;
    }

    // Used during emergencies when we need to replace the contract during an active auction
    function migrate(
        CancelledAuctionMap[] calldata _cancelledAuctions,
        BidHistoryMap[] calldata _bidHistory
    ) external onlyOwner {
        // Update all bid history.
        for (uint8 i = 0; i < _bidHistory.length; i++) {
            uint8 tokenId = _bidHistory[i].tokenId;
            for (uint256 j = 0; j < _bidHistory[i].bids.length; j++) {
                bidHistory[tokenId].push(
                    Bid(
                        _bidHistory[i].bids[j].walletBidder,
                        _bidHistory[i].bids[j].walletRecipient,
                        _bidHistory[i].bids[j].bidAmount
                    )
                );
            }
        }

        // Update all cancelled auctions.
        for (uint8 i = 0; i < _cancelledAuctions.length; i++) {
            uint8 tokenId = _cancelledAuctions[i].tokenId;
            cancelledAuctions[tokenId] = _cancelledAuctions[i].cancelled;
        }
    }
}

// File: Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: PrideIcons.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.11;

import "ERC721.sol";
import "Ownable.sol";
import "ECDSA.sol";

contract PrideIcons is ERC721, Ownable {
    using ECDSA for bytes32;

    uint8 public maxMintsPerTx;

    uint256 public silverPrice;
    uint256 public goldPrice;

    uint16 public silverSupply;
    uint16 public goldSupply;
    uint16 public platinumSupply;

    // Number of tokens minted - gold & silver
    // @VisibleForTesting
    uint16 internal mintedGold;
    uint16 internal mintedSilver;
    uint16 internal mintedPlatinum;

    // Keep track of platinum NFTs minted. This is useful for our Auction contract.
    mapping(uint16 => bool) public hasMintedPlatinum;

    // IPFS prefix URI for each token metadata.
    string private baseTokenURI;

    // IPFS URI for metadata that describes this contract.
    string private nftContractURI;

    // Those who bought during the Early Bird Sale.
    mapping(address => bool) public earlyBirdSaleList;

    // The amount each address bought during the Early Bird Sale.
    mapping(address => uint256) public earlyBirdSalePurchasesSilver;
    mapping(address => uint256) public earlyBirdSalePurchasesGold;

    // Those who bought during the presale.
    mapping(address => bool) public presaleList;

    // The amount each address bought during the presale.
    mapping(address => uint256) public presalePurchasesSilver;
    mapping(address => uint256) public presalePurchasesGold;

    // The amount each allowed address has reserved to purchase during the presale.
    // Allowed addresses are those who successfully registered before the presale.
    mapping(address => uint256) public presaleReservedSilver;
    mapping(address => uint256) public presaleReservedGold;

    // A salt to ensure transaction origin.
    mapping(bytes32 => bool) private nonces;

    // Provenance hash.
    string public proof;

    bool public earlyBirdSaleLive;
    bool public presaleLive;
    bool public publicSaleLive;

    bool public metadataLocked;
    bool public auctionLocked;

    uint16 public mintedSilverGifts;
    uint16 public mintedGoldGifts;

    uint256 public revealTimestamp;
    uint256 public startingIndexBlock;
    uint256 public startingIndex;

    // @VisibleForTesting
    address internal auctionAddress;
    // @VisibleForTesting
    address internal vaultAddress;
    // Address from the approved sender.
    address internal signerAddress;

    // Paper wallets
    mapping(address => bool) private paperWallets;

    constructor(
        address _vaultAddress,
        address _signerAddress,
        string memory _baseTokenURI,
        string memory _nftContractURI,
        uint8 _maxMintsPerTx,
        uint256 _silverPrice,
        uint256 _goldPrice,
        uint16 _silverSupply,
        uint16 _goldSupply,
        uint16 _platinumSupply,
        address[] memory _paperWallets
    ) ERC721("PrideIcons", "PI") {
        require(_vaultAddress != address(0), "NULL_ADDRESS");
        require(_signerAddress != address(0), "NULL_ADDRESS");
        vaultAddress = _vaultAddress;
        signerAddress = _signerAddress;
        baseTokenURI = _baseTokenURI;
        nftContractURI = _nftContractURI;
        maxMintsPerTx = _maxMintsPerTx;
        silverPrice = _silverPrice;
        goldPrice = _goldPrice;
        silverSupply = _silverSupply;
        goldSupply = _goldSupply;
        platinumSupply = _platinumSupply;
        initializePaperWallets(_paperWallets);
    }

    modifier metadataNotLocked() {
        require(!metadataLocked, "CONTRACT_METADATA_METHODS_ARE_LOCKED");
        _;
    }

    modifier auctionNotLocked() {
        require(!auctionLocked, "CONTRACT_LOCKED_DOWN");
        _;
    }

    modifier onlyPaperOrOwner() {
        require(
            paperWallets[msg.sender] || owner() == msg.sender,
            "ONLY_PAPER_WALLETS_OR_OWNER_ALLOWED"
        );
        _;
    }

    // ------------------------------------------------------
    // Mint functions
    // ------------------------------------------------------
    function mint(
        bytes32 hash,
        bytes calldata signature,
        bytes32 nonce,
        uint16 silverQuantityToMint,
        uint16 goldQuantityToMint
    ) external payable {
        require(
            publicSaleLive && !presaleLive && !earlyBirdSaleLive,
            "SALE_CLOSED"
        );
        require(
            mintedSilver + silverQuantityToMint <= silverSupply,
            "OUT_OF_STOCK_SILVER"
        );
        require(
            mintedGold + goldQuantityToMint <= goldSupply,
            "OUT_OF_STOCK_GOLD"
        );
        require(
            silverQuantityToMint + goldQuantityToMint <= maxMintsPerTx,
            "EXCEED_QUANTITY_LIMIT"
        );
        require(
            msg.value >=
                (silverPrice *
                    silverQuantityToMint +
                    goldPrice *
                    goldQuantityToMint),
            "INSUFFICIENT_ETH"
        );
        require(matchAddresSigner(hash, signature), "UNAUTHORIZED_MINT");
        require(!nonces[nonce], "HASH_USED");
        require(
            hashTransaction(
                msg.sender,
                silverQuantityToMint + goldQuantityToMint,
                nonce
            ) == hash,
            "HASH_FAIL"
        );

        if (silverQuantityToMint > 0) {
            mintSilverQuantity(msg.sender, silverQuantityToMint);
        }
        if (goldQuantityToMint > 0) {
            mintGoldQuantity(msg.sender, goldQuantityToMint);
        }

        nonces[nonce] = true;

        // If we're done, set the starting index of the collection.
        trySetStartingIndex();
    }

    function mintEarlyBird(
        uint16 silverQuantityToMint,
        uint16 goldQuantityToMint
    ) external payable {
        require(
            earlyBirdSaleLive && !publicSaleLive && !presaleLive,
            "EARLY_BIRD_SALE_CLOSED"
        );
        require(earlyBirdSaleList[msg.sender], "UNAUTHORIZED");
        require(
            mintedSilver + silverQuantityToMint <= silverSupply,
            "OUT_OF_STOCK_SILVER"
        );
        require(
            mintedGold + goldQuantityToMint <= goldSupply,
            "OUT_OF_STOCK_GOLD"
        );
        require(
            earlyBirdSalePurchasesSilver[msg.sender] +
                silverQuantityToMint +
                earlyBirdSalePurchasesGold[msg.sender] +
                goldQuantityToMint <=
                maxMintsPerTx,
            "EXCEED_EARLY_BIRD_SALE_PERSONAL_LIMIT"
        );
        require(
            msg.value >=
                (silverPrice *
                    silverQuantityToMint +
                    goldPrice *
                    goldQuantityToMint),
            "INSUFFICIENT_ETH"
        );

        earlyBirdSalePurchasesSilver[msg.sender] += silverQuantityToMint;
        earlyBirdSalePurchasesGold[msg.sender] += goldQuantityToMint;

        if (silverQuantityToMint > 0) {
            mintSilverQuantity(msg.sender, silverQuantityToMint);
        }
        if (goldQuantityToMint > 0) {
            mintGoldQuantity(msg.sender, goldQuantityToMint);
        }

        // If we sold everything during early bird sale, set the starting index of the collection.
        trySetStartingIndex();
    }

    function mintPresale(uint16 silverQuantityToMint, uint16 goldQuantityToMint)
        external
        payable
    {
        require(
            presaleLive && !publicSaleLive && !earlyBirdSaleLive,
            "PRESALE_CLOSED"
        );
        require(presaleList[msg.sender], "UNAUTHORIZED");
        require(
            mintedSilver + silverQuantityToMint <= silverSupply,
            "OUT_OF_STOCK_SILVER"
        );
        require(
            mintedGold + goldQuantityToMint <= goldSupply,
            "OUT_OF_STOCK_GOLD"
        );
        require(
            presaleReservedSilver[msg.sender] == silverQuantityToMint &&
                presaleReservedGold[msg.sender] == goldQuantityToMint,
            "QUANTITY_MUST_BE_EQUAL_TO_RESERVED"
        );
        require(
            presalePurchasesSilver[msg.sender] +
                silverQuantityToMint +
                presalePurchasesGold[msg.sender] +
                goldQuantityToMint <=
                maxMintsPerTx,
            "EXCEED_PRESALE_PERSONAL_LIMIT"
        );
        require(
            msg.value >=
                (silverPrice *
                    silverQuantityToMint +
                    goldPrice *
                    goldQuantityToMint),
            "INSUFFICIENT_ETH"
        );

        presalePurchasesSilver[msg.sender] += silverQuantityToMint;
        presalePurchasesGold[msg.sender] += goldQuantityToMint;

        if (silverQuantityToMint > 0) {
            mintSilverQuantity(msg.sender, silverQuantityToMint);
        }
        if (goldQuantityToMint > 0) {
            mintGoldQuantity(msg.sender, goldQuantityToMint);
        }

        // If we sold everything during presale, set the starting index of the collection.
        trySetStartingIndex();
    }

    function mintPlatinum(uint8 tokenId, address buyer)
        external
        auctionNotLocked
    {
        require(auctionAddress != address(0), "AUCTION_CONTRACT_UNINITIALIZED");
        require(buyer != address(0), "NULL_ADDRESS");
        require(msg.sender == auctionAddress, "UNAUTHORIZED");
        require(mintedPlatinum + 1 <= platinumSupply, "OUT_OF_STOCK");
        require(tokenId <= platinumSupply, "TOKEN_ID_OUT_OF_PLATINUM_RANGE");

        _safeMint(buyer, tokenId);
        mintedPlatinum++;
        hasMintedPlatinum[tokenId] = true;
    }

    function giftSilver(address[] calldata receivers) external onlyOwner {
        require(
            mintedSilver + receivers.length <= silverSupply,
            "OUT_OF_STOCK_SILVER"
        );

        // Safe conversion since silverSupply is less than 2^16.
        mintedSilverGifts += uint16(receivers.length);
        for (uint16 i = 0; i < receivers.length; i++) {
            _safeMint(receivers[i], getNextSilverTokenId() + i);
        }
        mintedSilver += uint16(receivers.length);
    }

    function giftGold(address[] calldata receivers) external onlyOwner {
        require(
            mintedGold + receivers.length <= goldSupply,
            "OUT_OF_STOCK_GOLD"
        );

        // Safe conversion since goldSupply is less than 2^16.
        mintedGoldGifts += uint16(receivers.length);
        for (uint16 i = 0; i < receivers.length; i++) {
            _safeMint(receivers[i], getNextGoldTokenId() + i);
        }
        mintedGold += uint16(receivers.length);
    }

    // ------------------------------------------------------
    // Paper Mint functions
    // ------------------------------------------------------

    function paperMintSilver(address userWallet, uint8 quantity)
        public
        payable
        onlyPaperOrOwner
    {
        string memory ineligibilityReason = getClaimIneligibilityReasonSilver(
            userWallet,
            quantity
        );
        require(
            keccak256(abi.encodePacked(ineligibilityReason)) ==
                keccak256(abi.encodePacked("")),
            ineligibilityReason
        );

        if (presaleLive) {
            presalePurchasesSilver[userWallet] += quantity;
        } else if (earlyBirdSaleLive) {
            earlyBirdSalePurchasesSilver[userWallet] += quantity;
        }

        // Mint.
        mintSilverQuantity(userWallet, quantity);

        // If we sold everything during presale, set the starting index of the collection.
        trySetStartingIndex();
    }

    function paperMintGold(address userWallet, uint8 quantity)
        public
        payable
        onlyPaperOrOwner
    {
        string memory ineligibilityReason = getClaimIneligibilityReasonGold(
            userWallet,
            quantity
        );
        require(
            keccak256(abi.encodePacked(ineligibilityReason)) ==
                keccak256(abi.encodePacked("")),
            ineligibilityReason
        );

        if (presaleLive) {
            presalePurchasesGold[userWallet] += quantity;
        } else if (earlyBirdSaleLive) {
            earlyBirdSalePurchasesGold[userWallet] += quantity;
        }

        // Mint.
        mintGoldQuantity(userWallet, quantity);

        // If we sold everything during presale, set the starting index of the collection.
        trySetStartingIndex();
    }

    function getClaimIneligibilityReasonSilver(
        address userWallet,
        uint8 quantity
    ) public view returns (string memory) {
        if (presaleLive) {
            return
                getClaimIneligibilityReason(
                    userWallet,
                    quantity,
                    silverPrice,
                    mintedSilver,
                    silverSupply,
                    presaleReservedSilver[userWallet],
                    presalePurchasesSilver[userWallet],
                    presalePurchasesGold[userWallet]
                );
        }

        if (earlyBirdSaleLive) {
            return
                getClaimIneligibilityReason(
                    userWallet,
                    quantity,
                    silverPrice,
                    mintedSilver,
                    silverSupply,
                    0,
                    earlyBirdSalePurchasesSilver[userWallet],
                    earlyBirdSalePurchasesGold[userWallet]
                );
        }

        return
            getClaimIneligibilityReason(
                userWallet,
                quantity,
                silverPrice,
                mintedSilver,
                silverSupply,
                0,
                0,
                0
            );
    }

    function getClaimIneligibilityReasonGold(address userWallet, uint8 quantity)
        public
        view
        returns (string memory)
    {
        if (presaleLive) {
            return
                getClaimIneligibilityReason(
                    userWallet,
                    quantity,
                    goldPrice,
                    mintedGold,
                    goldSupply,
                    presaleReservedGold[userWallet],
                    presalePurchasesSilver[userWallet],
                    presalePurchasesGold[userWallet]
                );
        }

        if (earlyBirdSaleLive) {
            return
                getClaimIneligibilityReason(
                    userWallet,
                    quantity,
                    goldPrice,
                    mintedGold,
                    goldSupply,
                    0,
                    earlyBirdSalePurchasesSilver[userWallet],
                    earlyBirdSalePurchasesGold[userWallet]
                );
        }

        return
            getClaimIneligibilityReason(
                userWallet,
                quantity,
                goldPrice,
                mintedGold,
                goldSupply,
                0,
                0,
                0
            );
    }

    function getClaimIneligibilityReason(
        address userWallet,
        uint8 quantity,
        uint256 price,
        uint16 purchased,
        uint16 supply,
        uint256 reserved,
        uint256 purchasedSilver,
        uint256 purchasedGold
    ) private view returns (string memory) {
        if (quantity == 0) {
            return "NO_QUANTITY";
        }

        if (msg.value < (quantity * price)) {
            return "INSUFFICIENT_ETH";
        }

        if (purchased + quantity > supply) {
            return "NOT_ENOUGH_SUPPLY";
        }

        if (purchasedSilver + purchasedGold + quantity > maxMintsPerTx) {
            return "EXCEED_LIMIT";
        }

        if (earlyBirdSaleLive) {
            if (!earlyBirdSaleList[userWallet]) {
                return "NOT_ON_ALLOWLIST";
            }
        } else if (presaleLive) {
            if (!presaleList[userWallet]) {
                return "NOT_ON_ALLOWLIST";
            }

            if (reserved != quantity) {
                return "QUANTITY_MUST_BE_EQUAL_TO_RESERVED";
            }
        } else if (!publicSaleLive) {
            return "NOT_LIVE";
        }

        return "";
    }

    // ------------------------------------------------------
    // Utility
    // ------------------------------------------------------

    function mintSilverQuantity(address wallet, uint16 quantityToMint) private {
        for (uint16 i = 0; i < quantityToMint; i++) {
            _safeMint(wallet, getNextSilverTokenId() + i);
        }
        mintedSilver += quantityToMint;
    }

    function mintGoldQuantity(address wallet, uint16 quantityToMint) private {
        for (uint16 i = 0; i < quantityToMint; i++) {
            _safeMint(wallet, getNextGoldTokenId() + i);
        }
        mintedGold += quantityToMint;
    }

    function getNextSilverTokenId() internal view returns (uint16) {
        return mintedSilver + goldSupply + platinumSupply + 1;
    }

    function getNextGoldTokenId() internal view returns (uint16) {
        return mintedGold + platinumSupply + 1;
    }

    /** Setting the starting index will reveal the mapping of all tokens in the collection. */
    function setStartingIndex() public {
        require(startingIndex == 0, "ALREADY_SET");
        require(startingIndexBlock != 0, "MISSING_SEED");
        setStartingIndexInternal();
    }

    /** If the time is right, set the starting index of our collection and record the block number used. */
    function trySetStartingIndex() private {
        if (
            startingIndexBlock == 0 &&
            ((mintedSilver + mintedGold == silverSupply + goldSupply) ||
                block.timestamp >= revealTimestamp)
        ) {
            startingIndexBlock = block.number;
            setStartingIndexInternal();
        }
    }

    function setStartingIndexInternal() private {
        if (block.number - startingIndexBlock > 255) {
            // The EVM only stores the last 256 block hashs.
            startingIndex =
                uint256(
                    keccak256(
                        abi.encodePacked(
                            blockhash(block.number - 1),
                            block.coinbase,
                            block.timestamp
                        )
                    )
                ) %
                goldSupply;
        } else {
            startingIndex =
                uint256(
                    keccak256(
                        abi.encodePacked(
                            blockhash(startingIndexBlock),
                            block.coinbase,
                            block.timestamp
                        )
                    )
                ) %
                goldSupply;
        }

        // Prevent the default sequence.
        if (startingIndex == 0) {
            startingIndex = (block.number + 1) % goldSupply;
        }
    }

    function hashTransaction(
        address sender,
        uint256 quantity,
        bytes32 nonce
    ) internal pure returns (bytes32) {
        bytes32 hash = keccak256(
            abi.encodePacked(
                "\x19Ethereum Signed Message:\n32",
                keccak256(abi.encodePacked(sender, quantity, nonce))
            )
        );

        return hash;
    }

    function matchAddresSigner(bytes32 hash, bytes calldata signature)
        internal
        view
        returns (bool)
    {
        return signerAddress == hash.recover(signature);
    }

    function _baseURI() internal view override(ERC721) returns (string memory) {
        return baseTokenURI;
    }

    // For OpenSea and other marketplaces.
    function contractURI() public view returns (string memory) {
        return nftContractURI;
    }

    // ------------------------------------------------------
    // Owner functions
    // ------------------------------------------------------
    function addToEarlyBirdSale(address[] calldata entries) external onlyOwner {
        for (uint256 i = 0; i < entries.length; i++) {
            address entry = entries[i];
            require(entry != address(0), "NULL_ADDRESS");
            require(!earlyBirdSaleList[entry], "DUPLICATE_ENTRY");

            earlyBirdSaleList[entry] = true;
        }
    }

    function addToPresale(
        address[] calldata entries,
        uint256[] calldata reservedPresaleSilver,
        uint256[] calldata reservedPresaleGold
    ) external onlyOwner {
        require(
            entries.length == reservedPresaleSilver.length &&
                reservedPresaleSilver.length == reservedPresaleGold.length,
            "UNEQUAL_LENGTH"
        );
        for (uint256 i = 0; i < entries.length; i++) {
            address entry = entries[i];
            uint256 silverQuantityToReserve = reservedPresaleSilver[i];
            uint256 goldQuantityToReserve = reservedPresaleGold[i];
            require(
                silverQuantityToReserve > 0 || goldQuantityToReserve > 0,
                "RESERVE_MUST_BE_GREATER_THAN_ZERO"
            );
            require(
                silverQuantityToReserve + goldQuantityToReserve <=
                    maxMintsPerTx,
                "RESERVE_OVER_LIMIT"
            );
            require(entry != address(0), "NULL_ADDRESS");
            require(!presaleList[entry], "DUPLICATE_ENTRY");

            presaleList[entry] = true;
            if (silverQuantityToReserve > 0) {
                presaleReservedSilver[entry] = silverQuantityToReserve;
            }
            if (goldQuantityToReserve > 0) {
                presaleReservedGold[entry] = goldQuantityToReserve;
            }
        }
    }

    function removeFromEarlyBirdSaleList(address[] calldata entries)
        external
        onlyOwner
    {
        for (uint256 i = 0; i < entries.length; i++) {
            address entry = entries[i];
            require(entry != address(0), "NULL_ADDRESS");

            earlyBirdSaleList[entry] = false;
        }
    }

    function removeFromPresaleList(address[] calldata entries)
        external
        onlyOwner
    {
        for (uint256 i = 0; i < entries.length; i++) {
            address entry = entries[i];
            require(entry != address(0), "NULL_ADDRESS");

            presaleList[entry] = false;
            presaleReservedSilver[entry] = 0;
            presaleReservedGold[entry] = 0;
        }
    }

    function modifyTokenPrices(uint256 newSilverPrice, uint256 newGoldPrice)
        external
        onlyOwner
    {
        silverPrice = newSilverPrice;
        goldPrice = newGoldPrice;
    }

    function setVaultAddress(address newVault) external onlyOwner {
        require(newVault != address(0), "NULL_ADDRESS");
        vaultAddress = newVault;
    }

    function lockMetadata() external onlyOwner {
        metadataLocked = true;
    }

    function lockAuction() external onlyOwner {
        auctionAddress = address(0);
        auctionLocked = true;
    }

    function unlockAuction(address newAuctionAddress) external onlyOwner {
        auctionLocked = false;
        auctionAddress = newAuctionAddress;
    }

    /** When live, disables all other sales. */
    function toggleEarlyBirdStatus() external onlyOwner {
        earlyBirdSaleLive = !earlyBirdSaleLive;
        if (earlyBirdSaleLive) {
            presaleLive = false;
            publicSaleLive = false;
        }
    }

    /** When live, disables all other sales. */
    function togglePresaleStatus() external onlyOwner {
        presaleLive = !presaleLive;
        if (presaleLive) {
            earlyBirdSaleLive = false;
            publicSaleLive = false;
        }
    }

    /** When live, disables all other sales. */
    function togglePublicSaleStatus() external onlyOwner {
        publicSaleLive = !publicSaleLive;
        if (publicSaleLive) {
            earlyBirdSaleLive = false;
            presaleLive = false;
        }
    }

    function setRevealTimestamp(uint256 timestamp) public onlyOwner {
        revealTimestamp = timestamp;
    }

    function setSignerAddress(address addressToSet) external onlyOwner {
        require(addressToSet != address(0), "NULL_ADDRESS");
        signerAddress = addressToSet;
    }

    function withdrawAll() external onlyOwner {
        payable(vaultAddress).transfer(address(this).balance);
    }

    function setContractURI(string calldata uri) external onlyOwner {
        nftContractURI = uri;
    }

    function assignAuctionContractAddress(address _auctionAddress)
        external
        onlyOwner
    {
        require(_auctionAddress != address(0), "NULL_ADDRESS");
        auctionAddress = _auctionAddress;
    }

    function adjustMaxMintsPerTx(uint8 _maxMintsPerTx) external onlyOwner {
        maxMintsPerTx = _maxMintsPerTx;
    }

    // virtual for tests
    function initializePaperWallets(address[] memory wallets)
        public
        virtual
        onlyOwner
    {
        for (uint8 i = 0; i < wallets.length; i++) {
            paperWallets[wallets[i]] = true;
        }
    }

    function adjustSupply(
        uint16 newSilverSupply,
        uint16 newGoldSupply,
        uint16 newPlatinumSupply
    ) external onlyOwner {
        require(
            newPlatinumSupply == platinumSupply ||
                (mintedPlatinum <= newPlatinumSupply &&
                    mintedSilver == 0 &&
                    mintedGold == 0),
            "SUPPLY_CANNOT_CHANGE_AFTER_MINT"
        );
        require(
            newGoldSupply == goldSupply ||
                (mintedGold <= newGoldSupply && mintedSilver == 0),
            "SUPPLY_CANNOT_CHANGE_AFTER_MINT"
        );
        require(
            mintedSilver <= newSilverSupply,
            "SUPPLY_CANNOT_CHANGE_AFTER_MINT"
        );

        silverSupply = newSilverSupply;
        goldSupply = newGoldSupply;
        platinumSupply = newPlatinumSupply;
    }

    // ------------------------------------------------------
    // NFT meta functions
    // ------------------------------------------------------
    function setBaseTokenURI(string calldata uri)
        external
        onlyOwner
        metadataNotLocked
    {
        baseTokenURI = uri;
    }

    function setProvenanceHash(string calldata hash)
        external
        onlyOwner
        metadataNotLocked
    {
        proof = hash;
    }

    /**
     * Set the starting index block for the collection if setting
     * starting index is blocked.
     */
    function emergencySetStartingIndexBlock()
        external
        onlyOwner
        metadataNotLocked
    {
        require(startingIndex == 0, "STARTING_INDEX_ALREADY_SET");
        startingIndexBlock = block.number;
    }
}

// File: ERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "IERC721.sol";
import "IERC721Receiver.sol";
import "IERC721Metadata.sol";
import "Address.sol";
import "Context.sol";
import "Strings.sol";
import "ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        require(operator != _msgSender(), "ERC721: approve to caller");

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer of token that is not own");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: IERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: IERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: IERC721Receiver.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: IERC721Metadata.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: Strings.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: ERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: ECDSA.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Elliptic Curve Digital Signature Algorithm (ECDSA) operations.
 *
 * These functions can be used to verify that a message was signed by the holder
 * of the private keys of a given address.
 */
library ECDSA {
    enum RecoverError {
        NoError,
        InvalidSignature,
        InvalidSignatureLength,
        InvalidSignatureS,
        InvalidSignatureV
    }

    function _throwError(RecoverError error) private pure {
        if (error == RecoverError.NoError) {
            return; // no error: do nothing
        } else if (error == RecoverError.InvalidSignature) {
            revert("ECDSA: invalid signature");
        } else if (error == RecoverError.InvalidSignatureLength) {
            revert("ECDSA: invalid signature length");
        } else if (error == RecoverError.InvalidSignatureS) {
            revert("ECDSA: invalid signature 's' value");
        } else if (error == RecoverError.InvalidSignatureV) {
            revert("ECDSA: invalid signature 'v' value");
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature` or error string. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     *
     * Documentation for signature generation:
     * - with https://web3js.readthedocs.io/en/v1.3.4/web3-eth-accounts.html#sign[Web3.js]
     * - with https://docs.ethers.io/v5/api/signer/#Signer-signMessage[ethers]
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, bytes memory signature) internal pure returns (address, RecoverError) {
        // Check the signature length
        // - case 65: r,s,v signature (standard)
        // - case 64: r,vs signature (cf https://eips.ethereum.org/EIPS/eip-2098) _Available since v4.1._
        if (signature.length == 65) {
            bytes32 r;
            bytes32 s;
            uint8 v;
            // ecrecover takes the signature parameters, and the only way to get them
            // currently is to use assembly.
            assembly {
                r := mload(add(signature, 0x20))
                s := mload(add(signature, 0x40))
                v := byte(0, mload(add(signature, 0x60)))
            }
            return tryRecover(hash, v, r, s);
        } else if (signature.length == 64) {
            bytes32 r;
            bytes32 vs;
            // ecrecover takes the signature parameters, and the only way to get them
            // currently is to use assembly.
            assembly {
                r := mload(add(signature, 0x20))
                vs := mload(add(signature, 0x40))
            }
            return tryRecover(hash, r, vs);
        } else {
            return (address(0), RecoverError.InvalidSignatureLength);
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature`. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     */
    function recover(bytes32 hash, bytes memory signature) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, signature);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `r` and `vs` short-signature fields separately.
     *
     * See https://eips.ethereum.org/EIPS/eip-2098[EIP-2098 short signatures]
     *
     * _Available since v4.3._
     */
    function tryRecover(
        bytes32 hash,
        bytes32 r,
        bytes32 vs
    ) internal pure returns (address, RecoverError) {
        bytes32 s;
        uint8 v;
        assembly {
            s := and(vs, 0x7fffffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff)
            v := add(shr(255, vs), 27)
        }
        return tryRecover(hash, v, r, s);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `r and `vs` short-signature fields separately.
     *
     * _Available since v4.2._
     */
    function recover(
        bytes32 hash,
        bytes32 r,
        bytes32 vs
    ) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, r, vs);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `v`,
     * `r` and `s` signature fields separately.
     *
     * _Available since v4.3._
     */
    function tryRecover(
        bytes32 hash,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal pure returns (address, RecoverError) {
        // EIP-2 still allows signature malleability for ecrecover(). Remove this possibility and make the signature
        // unique. Appendix F in the Ethereum Yellow paper (https://ethereum.github.io/yellowpaper/paper.pdf), defines
        // the valid range for s in (301): 0 < s < secp256k1n ÷ 2 + 1, and for v in (302): v ∈ {27, 28}. Most
        // signatures from current libraries generate a unique signature with an s-value in the lower half order.
        //
        // If your library generates malleable signatures, such as s-values in the upper range, calculate a new s-value
        // with 0xFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFEBAAEDCE6AF48A03BBFD25E8CD0364141 - s1 and flip v from 27 to 28 or
        // vice versa. If your library also generates signatures with 0/1 for v instead 27/28, add 27 to v to accept
        // these malleable signatures as well.
        if (uint256(s) > 0x7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF5D576E7357A4501DDFE92F46681B20A0) {
            return (address(0), RecoverError.InvalidSignatureS);
        }
        if (v != 27 && v != 28) {
            return (address(0), RecoverError.InvalidSignatureV);
        }

        // If the signature is valid (and not malleable), return the signer address
        address signer = ecrecover(hash, v, r, s);
        if (signer == address(0)) {
            return (address(0), RecoverError.InvalidSignature);
        }

        return (signer, RecoverError.NoError);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `v`,
     * `r` and `s` signature fields separately.
     */
    function recover(
        bytes32 hash,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, v, r, s);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from a `hash`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes32 hash) internal pure returns (bytes32) {
        // 32 is the length in bytes of hash,
        // enforced by the type signature above
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n32", hash));
    }

    /**
     * @dev Returns an Ethereum Signed Typed Data, created from a
     * `domainSeparator` and a `structHash`. This produces hash corresponding
     * to the one signed with the
     * https://eips.ethereum.org/EIPS/eip-712[`eth_signTypedData`]
     * JSON-RPC method as part of EIP-712.
     *
     * See {recover}.
     */
    function toTypedDataHash(bytes32 domainSeparator, bytes32 structHash) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19\x01", domainSeparator, structHash));
    }
}